

Arquitetura e Administração de Bases de Dados  
Oracle10g PL/SQL Programming

João Costa  
jcosta@isec.pt

## Agenda

---

- ▶ **Triggers**
  - ▶ De linha (ou de registo) – **ROW TRIGGER**
  - ▶ De comando – **STATEMENT TRIGGER**

## Triggers

- ▶ Semelhantes a procedimentos e funções
- ▶ Executam automaticamente em resposta a um evento
  - ▶ Comandos **DML**
  - ▶ Comandos **DDL**
  - ▶ Eventos da base de dados
- ▶ Os *triggers* **DML** estão associados a uma tabela ou vista

## Triggers

- ▶ Os *triggers* executam automaticamente com base num evento DML ou de sistema
- ▶ O tempo de execução pode ser:
  - ▶ Antes (**BEFORE**) ou depois (**AFTER**) do evento
  - ▶ Ao nível da linha (**ROW**) ou do comando (**STATEMENT**)
- ▶ A cláusula **WHEN** permite processamento condicional do *trigger*

## Criar um *trigger* DML

```
CREATE [OR REPLACE] TRIGGER nome_trigger
[ BEFORE | AFTER ]
[ INSERT | DELETE |
  UPDATE | UPDATE of colunas ]
ON nome_tabela
[ REFERENCING OLD as nome | NEW as nome ]
[ FOR EACH ROW ]
[ WHEN condições ]
DECLARE
  -- declaração de variáveis
BEGIN
  -- Código
END;
```

► 5

2021/22 - LEI - AABD - PL/SQL

## *Trigger* DML Ex. BEFORE ... FOR EACH ROW

```
CREATE OR REPLACE TRIGGER trig_ins_venda
  BEFORE INSERT ON vendas
  FOR EACH ROW
  WHEN (new.quantidade <= 0)
DECLARE
BEGIN
  RAISE_APPLICATION_ERROR(-20000,'Quantidade inválida');
END;
/
```

```
insert into vendas values (99,sysdate,1,1,10,-1,0);
```

```
Error report -
ORA-20000: Quantidade inválida
ORA-06512: na "LEIBD2C30.TRIG_INS_VENDA", linha 3
ORA-04088: erro durante a execução do trigger 'TRIG_INS_VENDA'
```

► 6

2021/22 - LEI - AABD - PL/SQL

## Tempo de execução

- ▶ Quando executar o código do trigger ?
  - ▶ Antes (**BEFORE**) ou
  - ▶ depois (**AFTER**) de acontecer o evento
- ▶ Scope de execução do código ?
  - ▶ Para cada linha afetada (**ROW**) ou
  - ▶ Por cada comando executado(**STATEMENT**)
- ▶ A cláusula **WHEN** permite processamento condicional

▶ 7

2021/22 - LEI - AABD - PL/SQL

## Eventos

- ▶ **INSERT, UPDATE, DELETE**
  - ▶ Usar o operador **OR** para incluir mais do que um evento. ex.:
    - ▶ **INSERT OR DELETE**
    - ▶ **INSERT OR DELETE OR UPDATE**
- ▶ No evento de **UPDATE**, a opção **OF nome\_da\_coluna**
  - ▶ **UPDATE OF** *coluna1, coluna2, ...*
  - ▶ *Permite indicar apenas quais as colunas que, se os valores forem alterados, permitem ativar o trigger*
- ▶ **ON nome\_da\_tabela**
  - ▶ *Apenas o nome de uma única tabela.*
  - ▶ Ex. **ON** *autores*

▶ 8

2021/22 - LEI - AABD - PL/SQL

## Corpo do *trigger*

- ▶ É um bloco PL/SQL
  - ▶ Deve incluir uma secção DECLARE
    - ▶ Se forem necessárias declarações
  - ▶ Pode referenciar identificadores de correlação (se row trigger)
    - ▶ Precedidos por ":" (ex: :NEW , :OLD)
  - ▶ Pode chamar outros subprogramas

▶ 9

2021/22 - LEI - AABD - PL/SQL

## Identificadores de correlação

- ▶ Variáveis especiais associadas a DML
  - ▶ Por defeito: OLD e NEW

DML Event	OLD Identifier	NEW Identifier
INSERT	---	Contains insert values
UPDATE	Contains values of the original row	Contains new value for any columns updated and original values for any columns not updated
DELETE	Contains values of the original row	Not Available (Note: "Not Available" indicates any references would retrieve a NULL value)

▶ 10

2021/22 - LEI - AABD - PL/SQL

## Trigger DML Ex. BEFORE ... FOR EACH ROW

```
CREATE OR REPLACE TRIGGER trig_ins_venda
  BEFORE INSERT ON vendas
  FOR EACH ROW
  WHEN (new.quantidade <= 0)
DECLARE
BEGIN
  RAISE_APPLICATION_ERROR(-20000,'Quantidade inválida');
END;
/
```

```
insert into vendas values (99,sysdate,1,1,10,-1,0);
```

```
Error report -
ORA-20000: Quantidade inválida
ORA-06512: na "LEIBD2C30.TRIG_INS_VENDA", linha 3
ORA-04088: erro durante a execução do trigger 'TRIG_INS_VENDA'
```

► 11

2021/22 - LEI - AABD - PL/SQL

## Trigger DML Ex. BEFORE ... FOR EACH ROW

```
CREATE OR REPLACE TRIGGER trig_ins_venda2
  BEFORE INSERT ON vendas
  FOR EACH ROW
DECLARE
  CURSOR C1 IS
    SELECT QUANT_EM_STOCK as STOCK
    FROM LIVROS
    WHERE CODIGO_LIVRO = :NEW.CODIGO_LIVRO;
BEGIN
  FOR R IN C1
  LOOP
    IF R.STOCK < :NEW.QUANTIDADE THEN
      RAISE_APPLICATION_ERROR(-20001,'STOCK INSUFICIENTE.MAX='||R.STOCK);
    END IF;
  END LOOP;
END;
/

insert into vendas values (99,sysdate,1,1,10,999999,0);
```

► 12

2021/22 - LEI - AABD - PL/SQL

## Trigger DML Ex. AFTER ... FOR EACH ROW

```
CREATE OR REPLACE TRIGGER trig_ins_venda3
  AFTER INSERT ON vendas
  FOR EACH ROW
BEGIN
  UPDATE LIVROS
  SET QUANT_EM_STOCK = QUANT_EM_STOCK - :NEW.QUANTIDADE
  WHERE CODIGO_LIVRO = :NEW.CODIGO_LIVRO;
END;
```

```
SELECT QUANT_EM_STOCK FROM LIVROS WHERE CODIGO_LIVRO=1;
QUANT_EM_STOCK
-----
        20

insert into vendas values (99,sysdate,1,1,10,3,0);

SELECT QUANT_EM_STOCK FROM LIVROS WHERE CODIGO_LIVRO=1;
QUANT_EM_STOCK
-----
        17
```

► 13

2021/22 - LEI - AABD - PL/SQL

## Trigger DML Ex. AFTER ... (COMANDO)

```
CREATE OR REPLACE TRIGGER trig_ins_venda3
  AFTER INSERT ON vendas

BEGIN
  UPDATE LIVROS
  SET QUANT_EM_STOCK = QUANT_EM_STOCK - :NEW.QUANTIDADE
  WHERE CODIGO_LIVRO = :NEW.CODIGO_LIVRO;
END;
```

```
Error report -
ORA-04082: referências NEW ou OLD não são permitidas em triggers a nível de
tabela
04082. 00000 - "NEW or OLD references not allowed in table level triggers"
*Cause:      The trigger is accessing "new" or "old" values in a table trigger.
*Action:      Remove any new or old references.
```

► 14

2021/22 - LEI - AABD - PL/SQL

## Trigger DML Ex. AFTER ... (COMANDO)

```
CREATE OR REPLACE TRIGGER trig_ins_venda4
  AFTER INSERT ON vendas

BEGIN
  UPDATE LIVROS L
  SET QUANT_EM_STOCK = 1000 - (SELECT SUM(QUANTIDADE)
                                FROM VENDAS V
                                WHERE V.CODIGO_LIVRO = L.CODIGO_LIVRO);
END;
```

- Não é possível utilizar o :NEW e/ou :OLD para saber qual informação do registo que ativou a execução do trigger.
- Não se sabe qual o registo.
- Apenas se sabe qual o tipo de comando.

▶ 15

2021/22 - LEI - AABD - PL/SQL

## Predicados condicionais

- ▶ **IF INSERTING**
- ▶ **IF DELETING**
- ▶ **IF UPDATING**
- ▶ Permite diferente processamento para cada tipo de comando DML
  - ▶ Múltiplos comandos DML podem disparar o *trigger*
- ▶ Pode especificar uma coluna
  - ▶ IF UPDATING('lastname') THEN...

▶ 16

2021/22 - LEI - AABD - PL/SQL



## Aplicações dos *triggers*

- ▶ **Auditing**
  - ▶ Registo de atividade na base de dados
  - ▶ Ex: Alterações de dados sensíveis (salários)
    - ▶ Usar um *trigger*
      - Guardar os valores originais e os novos numa tabela de registo
  - ▶ Havendo problemas no futuro existe registo da alteração

▶ 17

2021/22 - LEI - AABD - PL/SQL

## Aplicações dos *triggers*

- ▶ **Integridade dos dados**
  - ▶ Validações complexas ou impossíveis de fazer com restrições **CHECK**
  - ▶ Ex:  
Garantir que o preço não pode ser alterado para um valor inferior ao atual
    - ▶ Os valores de **NEW** e **OLD** podem ser comparados num *trigger*

▶ 18

2021/22 - LEI - AABD - PL/SQL

## Aplicações dos *triggers*

- ▶ Integridade referencial
  - ▶ Em restrições de chave forasteira
    - ▶ Se tentarmos alterar o valor da chave na “tabela “mãe”
      - Ocorre um erro se existirem registos associados na tabela “filha”
  - ▶ Um *trigger* pode fazer o *update* em cascata dos registos da tabela “filha”

▶ 19

2021/22 - LEI - AABD - PL/SQL

## Aplicações dos *triggers*

- ▶ Dados calculados
  - ▶ Colunas cujo valor seja calculado a partir dos valores de outras colunas
    - ▶ Ex: Tabela-resumo com vendas totais por produto
      - Esta tabela tem que ser atualizada em tempo real
    - Por cada nova venda registada
      - É disparado um *trigger* que actualiza o total de vendas do produto respetivo na tabela-resumo
      - É calculado o total da venda

▶ 20

2021/22 - LEI - AABD - PL/SQL

## Restrições dos *triggers*

- ▶ Não podem incluir comandos de controlo de transações
- ▶ Não podem usar os tipos de dados `LONG` e `LONG RAW`
- ▶ Erro “*mutating table*”
  - ▶ Tentar, num *trigger* de linha, modificar uma tabela que está a ser alterada pelo evento que disparou o *trigger*

▶ 21

2021/22 - LEI - AABD - PL/SQL

## Comando ALTER TRIGGER

- ▶ Usado para compilar e ativar/desativar o *trigger*

```
ALTER TRIGGER trigger_name COMPILE;
```

```
ALTER TRIGGER trigger_name DISABLE | ENABLE;
```

```
ALTER TABLE table_name DISABLE|ENABLE ALL TRIGGERS;
```

▶ 22

2021/22 - LEI - AABD - PL/SQL

## Apagar um *trigger*

```
DROP TRIGGER trigger_name;
```

### **Nota:**

Se uma tabela ou vista for apagada, todos os *triggers* DML a ela associados serão apagados

▶ 23

2021/22 - LEI - AABD - PL/SQL

## Dicionário de dados

- ▶ Igual aos outros subprogramas
  - ▶ Excepto para ver o código fonte
    - ▶ Vista **USER\_TRIGGERS**
      - Coluna Description – Código do cabeçalho
      - Coluna Trigger\_body – Código do corpo do *trigger*

▶ 24

2021/22 - LEI - AABD - PL/SQL

## Triggers - Resumo

- ▶ Os *triggers* executam automaticamente com base num evento DML ou de sistema
- ▶ O tempo de execução pode ser:
  - ▶ Antes (**BEFORE**) ou depois (**AFTER**) do evento
  - ▶ Ao nível da linha (**ROW**) ou do comando (**STATEMENT**)
- ▶ A cláusula **WHEN** permite processamento condicional do *trigger*

▶ 25

2021/22 - LEI - AABD - PL/SQL

## Triggers – Resumo (cont.)

- ▶ Identificadores de correlação permitem aceder aos valores envolvidos no comando DML
  - ▶ **NEW**, **OLD**
- ▶ Predicados condicionais permitem um processamento diferente para cada comando DML
  - ▶ **INSERTING**, **UPDATING**, **DELETING**

▶ 26

2021/22 - LEI - AABD - PL/SQL

## Exercícios

- ▶ Crie uma tabela **caixa** que para cada dia, registre a quantidade de livros vendidos, e o total recebido
- ▶ Implemente um **trigger** de comando que mantenha a tabela caixa atualizada
- ▶ Implemente um **trigger** de linha que mantenha a tabela caixa atualizada

▶ 27

2021/22 - LEI - AABD - PL/SQL

## Exercícios

- ▶ Implemente um **trigger** que ao remover um cliente da tabela **clientes**, apague da tabela **vendas** todos os livros adquiridos pelo cliente removido.
- ▶ À medida que os livros vão sendo vendidos, o atributo **quant\_em\_stock** da tabela livros necessita ser atualizado em conformidade. Implemente um **trigger** que mantenha atualizado o atributo **quant\_em\_stock**.
- ▶ Implemente um **trigger** que garanta que o nome do autor é inserido em maiúsculas

▶ 28

2021/22 - LEI - AABD - PL/SQL

## Triggers INSTEAD OF

- ▶ Permite contornar o problema das vistas “não modificáveis”
  - ▶ Um comando DML a uma vista dispara um *trigger* INSTEAD OF
- ▶ Os comandos DML do *trigger* serão executados nas tabelas base da vista
  - ▶ Usando valores do evento que disparou o *trigger*

▶ 29

2021/22 - LEI - AABD - PL/SQL

## INSTEAD OF - Exemplo

```
CREATE VIEW vendas_hoje AS
  SELECT  codigo_venda, codigo_livro, nome,
          quantidade, preco_unitario
  FROM    VENDAS v, clientes c
  WHERE   TO_CHAR(DATA_VENDA,'dd-mm-yyyy') =
          TO_CHAR(SYSDATE  ,'dd-mm-yyyy')
  AND     c.codigo_cliente = v.codigo_cliente;
```

```
insert into vendas_hoje values (99,1,'Carlos Mota',10,3);
```

Error report -

SQL Error: ORA-01776: não é possível modificar mais de uma tabela base através de uma vista de junção

01776. - "cannot modify more than one base table through a join view"  
 \*Cause: Columns belonging to more than one underlying table were either inserted into or updated.

\*Action: Phrase the statement as two or more separate statements.

▶ 30

2021/22 - LEI - AABD - PL/SQL

## INSTEAD OF - Exemplo

```
CREATE OR REPLACE TRIGGER trig_vendas_hoje
  INSTEAD OF INSERT ON vendas_hoje
  FOR EACH ROW
DECLARE
  codcliente CLIENTES.CODIGO_CLIENTE%TYPE;
BEGIN
  SELECT codigo_cliente INTO codcliente
  FROM   clientes
  WHERE  UPPER(nome) = UPPER(:NEW.nome);

  INSERT INTO vendas VALUES ( :new.codigo_venda,
                               SYSDATE, codcliente, :new.codigo_livro,
                               :new.quantidade,   :new.preco_unitario,
                               :new.quantidade * :new.preco_unitario );
END;
/
```

▶ 31

2021/22 - LEI - AABD - PL/SQL

## Triggers – Resumo (cont.)

- ▶ *Triggers* INSTEAD OF fornecem um mecanismo para alterar vistas “não modificáveis”
- ▶ O comando ALTER TRIGGER permite compilar e ativar/desativar um *trigger*
- ▶ A vista USER\_TRIGGERS, do dicionário de dados, permite ver o código dos *triggers*

▶ 32

2021/22 - LEI - AABD - PL/SQL